Nonlinear Dynamic Stochastic General Equilibrium models in Stata 16

David Schenck

Senior Econometrician
Stata

October 08, 2019

Introduction

- Dynamic stochastic general equilibrium
- Models linking observed time-series to unobserved shocks through a structure
- Methods for taking macroeconomic models to the data

Stata tools for analyzing DSGE models

- Estimation commands:
  - `dsge`, for DSGE models that are already linear in variables
  - `dsgenl`, in which we linearize the nonlinear model for you
- postestimation commands:
  - `irf`
  - `estat policy`
  - `estat transition`
  - `estat steady`
  - `estat stable`
  - `estat covariance`
  - Along with the usual `predict`, `test`, `forecast`, etc

Let's work an example

- Write down a model
- Estimate its parameters
- Investigate model properties with postestimation tools

Here's a model

- Households demand output, given inflation and interest rates:

$$1 = \beta E_t \left[\left(\frac{X_{t+1}}{X_t} \right)^{-1} \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

- Firms set prices, given output demand:

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t \left[\Pi_{t+1} - 1 \right]$$

- Central bank sets interest rate, given inflation

$$\beta R_t = \Pi_t^{1/\beta} M_t$$

Here's a model

- The model's control variables are determined by equations:

$$1 = \beta E_t \left[\left(\frac{X_{t+1}}{X_t} \right)^{-1} \left(\frac{R_t}{\Pi_{t+1} Z_{t+1}} \right) \right]$$

$$\phi + (\Pi_t - 1) = \frac{1}{\phi} X_t + \beta E_t \left[\Pi_{t+1} - 1 \right]$$

$$\beta R_t = \Pi_t^{1/\beta} M_t$$

- The model is completed by adding equations for the state variables:

$$\ln(Z_{t+1}) = \rho_z \ln(Z_t) + \xi_{t+1}$$

$$\ln(M_{t+1}) = \rho_m \ln(M_t) + e_{t+1}$$

Here's a model in Stata

```
. dsgenl  (1 = {beta}*(F.x/x)^(-1)*(r/(F.p*F.z))   )    ///
          ({phi}+(p-1) = 1/{phi}*x + {beta}*(F.p-1))   ///
          ({beta}*r = p^(1/{beta})*m                )    ///
          (ln(F.m) = {rhom}*ln(m))                       ///
          (ln(F.z) = {rhoz}*ln(z))                       ///
          , exostate(z m) observed(p r) unobserved(x)
```

Parameter estimation

```
. webuse usmacro2

. dsgenl  (1 = {beta}*(F.x/x)^(-1)*(r/(F.p*F.z)))       ///
>         ({phi}+(p-1) = 1/{phi}*x + {beta}*(F.p-1))    ///
>         ({beta}*r = p^(1/{beta})*m)                   ///
>         (ln(F.m) = {rhom}*ln(m))                      ///
>         (ln(F.z) = {rhoz}*ln(z))                      ///
>         , exostate(z m) observed(p r) unobserved(x)
Solving at initial parameter vector ...
Checking identification ...

First-order DSGE model

Sample: 1955q1 - 2015q4                         Number of obs     =        244
Log likelihood = -753.57131
```

	Coef.	OIM Std. Err.	z	P>|z|	[95% Conf.	Interval]
/structural						
beta	.5146672	.0783493	6.57	0.000	.3611054	.668229
phi	.1659058	.0474002	3.50	0.000	.0730032	.2588083
rhom	.7005483	.0452634	15.48	0.000	.6118335	.789263
rhoz	.9545256	.0186417	51.20	0.000	.9179886	.9910627
sd(e.z)	.650712	.1123897			.4304321	.8709918
sd(e.m)	2.318204	.3047452			1.720914	2.915493

Policy questions

What is the effect of an unexpected increase in interest rates?

Estimated DSGE model provides an answer to this question. We can subject the model to a shock, then see how that shock feeds through the rest of the system.

Effect on impact

```
. estat policy
Policy matrix
```

	Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
x						
z	2.59502	.9077695	2.86	0.004	.8158242	4.374215
m	-1.608216	.4049684	-3.97	0.000	-2.401939	-.8144921
p						
z	.8462697	.2344472	3.61	0.000	.3867617	1.305778
m	-.4172522	.0393623	-10.60	0.000	-.4944008	-.3401035
r						
z	1.644305	.2357604	6.97	0.000	1.182223	2.106387
m	.1892777	.0591622	3.20	0.001	.0733219	.3052335

Effect over time: impulse response functions

```
. irf set nkirf.irf, replace
. irf create model1
. irf graph irf, impulse(m) response(p x r m) byopts(yrescale) yline(0)
```

Impulse responses from the estimated model



Analyzing nonlinear DSGE models

- Every model has a structure and a reduced form
- `estat` commands allow you to explore features of the reduced form
  - long-run steady-state
  - policy matrix of control variables
  - transition matrix of state variables
  - impulse response functions
  - model-implied covariances

Some notation

- Models have a structure that link control variables $\mathbf{y}_t$ to latent state variables $\mathbf{x}_t$ and, ultimately, to shocks $\mathbf{e}_t$
- The structure comes from economic theory
- The reduced form looks like

$$
\begin{aligned}
\mathbf{y}_t &= \mathbf{g}(\mathbf{x}_t) \\
\mathbf{x}_{t+1} &= \mathbf{h}(\mathbf{x}_t) + \mathbf{e}_{t+1}
\end{aligned}
$$

Linear approximation

- Models have a structure that link control variables $\mathbf{y}_t$ to latent state variables $\mathbf{x}_t$ and, ultimately, to shocks $\mathbf{e}_t$
- The structure comes from economic theory
- The reduced form looks like

$$\mathbf{y}_t - \overline{\mathbf{y}} = \mathbf{G} \cdot (\mathbf{x}_t - \overline{\mathbf{x}})$$
$$\mathbf{x}_{t+1} - \overline{\mathbf{x}} = \mathbf{H} \cdot (\mathbf{x}_t - \overline{\mathbf{x}}) + \mathbf{e}_{t+1}$$

Linear approximation

- Linearized reduced form:

$$\mathbf{y}_t - \overline{\mathbf{y}} = \mathbf{G} \cdot (\mathbf{x}_t - \overline{\mathbf{x}})$$
$$\mathbf{x}_{t+1} - \overline{\mathbf{x}} = \mathbf{H} \cdot (\mathbf{x}_t - \overline{\mathbf{x}}) + \mathbf{e}_{t+1}$$

- $(\overline{\mathbf{x}}, \overline{\mathbf{y}})$: steady-state
- $\mathbf{G}$: policy matrix
- $\mathbf{H}$: transition matrix
- Stata supports both linear and log–linear approximations

The toolkit in action

- Set up the neoclassical growth model
- Fix some parameters at calibrated values; estimate others
- Explore the full set of model features

The stochastic growth model

$$1 = \beta E_t \left[\left(\frac{c_{t+1}}{c_t} \right)^{-1} (1 + r_{t+1} - \delta) \right] \qquad \text{(Consumption)}$$

$$y_t = z_t k_t^{\alpha} \qquad \text{(Output)}$$

$$r_t = \alpha z_t k_t^{\alpha - 1} \qquad \text{(Interest rate)}$$

$$k_{t+1} = y_t - c_t + (1 - \delta) k_t \qquad \text{(Capital accumulation)}$$

$$\ln z_{t+1} = \rho \ln z_t + e_{t+1} \qquad \text{(Productivity)}$$

The stochastic growth model in Stata

```
. dsgenl (1        = {beta}*(c/F.c)*(1+F.r-{delta}) )                 ///
>        (r        = {alpha}*y/k                     )                 ///
>        (y        = z*k^{alpha}                     )                 ///
>        (F.k      = y - c + (1-{delta})*k           )                 ///
>        (ln(F.z) = {rhoz}*ln(z)                     ),                ///
>        exostate(z) endostate(k) observed(y) unobserved(c r)
```

Data

```
. import fred GDPC1
. generate dateq = qofd(daten)
. tsset dateq, quarterly
. generate lgdp = 100*ln(GDPC1)
. tsfilter hp y = lgdp
```

Data



Filtered real GDP
Percent deviation from trend
4
2
0
−2
−4
−6
1950q1
1960q1
1970q1
1980q1
1990q1
2000q1
2010q1
2020q1
dateq

Parameter estimation

```
. constraint 1 _b[beta]=0.96

. constraint 2 _b[alpha]=0.36

. constraint 3 _b[delta]=0.025

. dsgenl (1       = {beta}*(c/F.c)*(1+F.r-{delta}) )                ///
>        (r       = {alpha}*y/k                     )                ///
>        (y       = z*k^{alpha}                     )                ///
>        (F.k     = y - c + (1-{delta})*k           )                ///
>        (ln(F.z) = {rhoz}*ln(z)                    ),               ///
>        constraint(1/3) nocnsreport nolog                           ///
>        exostate(z) endostate(k) observed(y) unobserved(c r)
Solving at initial parameter vector ...
Checking identification ...

First-order DSGE model

Sample: 1947q1 - 2019q1                         Number of obs     =        289
Log likelihood = -362.93403
------------------------------------------------------------------------------
             |                 OIM
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
/structural  |
        beta |        .96  (constrained)
       delta |       .025  (constrained)
       alpha |        .36  (constrained)
        rhoz |   .8391786   .0325307    25.80   0.000     .7754197    .9029375
-------------+----------------------------------------------------------------
      sd(e.z)|   .8470234   .0352336                      .7779668      .91608
------------------------------------------------------------------------------
```

Steady–state

The place to which model variables come to rest after all shocks work their way through the system

```
. estat steady

Location of model steady-state
```

	Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
k	13.94329	.	.	.	.	.
z	1	.	.	.	.	.
c	2.233508	.	.	.	.	.
r	.0666667	.	.	.	.	.
y	2.582091	.	.	.	.	.

Note: Standard errors reported as missing for constrained steady-state values.

Steady–state: behind the scenes

- recall consumption equation:

$$1 = \beta E_t \left[\left(\frac{c_{t+1}}{c_t} \right)^{-1} (1 + r_{t+1} - \delta) \right]$$

- In steady-state:

$$\bar{r} = \frac{1}{\beta} - 1 + \delta$$

- which delivers the value $r = 0.06667$ in the table

Policy matrix

The effect on impact of a change in the state variables on the control variables

```
. estat policy
Policy matrix
```

		Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
c							
	k	.6371815	.	.	.	.	.
	z	.266745	.0244774	10.90	0.000	.2187701	.3147198
r							
	k	-.64	.	.	.	.	.
	z	1	.	.	.	.	.
y							
	k	.36	.	.	.	.	.
	z	1	.	.	.	.	.

Note: Standard errors reported as missing for constrained policy matrix values.

Policy matrix: behind the scenes

- Recall output equation:

$$y_t = z_t k_t^{\alpha}$$

- Loglinear output equation

$$\hat{y}_t = \hat{z}_t + \alpha \hat{k}_t$$

- which delivers the coefficients 1 and α seen in the table

State transition matrix

The dynamics of state variables through time

```
. estat transition
```

Transition matrix of state variables

	Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
F.k						
k	.9395996	.	.	.	.	.
z	.1424566	.0039209	36.33	0.000	.1347717	.1501414
F.z						
k	0	(omitted)				
z	.8391786	.0325307	25.80	0.000	.7754197	.9029375

Note: Standard errors reported as missing for constrained transition matrix values.

State transition matrix: behind the scenes

- Productivity equation:

$$\ln z_{t+1} = \rho \ln z_t + e_{t+1}$$

- Loglinear productivity equation:

$$\hat{z}_{t+1} = \rho \hat{z}_t + \hat{e}_{t+1}$$

- which delivers the coefficient $\hat{\rho}$ seen in the table

Model–implied covariances

```
. estat covariance y

Estimated covariances of model variables
```

	Coef.	Delta-method Std. Err.	z	P>|z|	[95% Conf.	Interval]
y						
var(y)	3.872087	.9694708	3.99	0.000	1.971959	5.772215

Impulse responses

The effect of a shock on the model variables through time
Setup:

```
. irf set stochirf.irf, replace
. irf create stochastic_model, step(40)
. irf graph irf, impulse(z) response(y c k z) yline(0) xlabel(0(4)40)
```

Impulse responses

Image:



Conclusion

- `dsgenl` estimates the parameters of nonlinear DSGE models
- View steady–state, policy matrix, transition matrix
- View model–implied covariances
- Create and analyze impulse responses

Thank You!